Credit Default Options with Mathematica 10

Credit default options represent option products on the forward CDS rate and are credit analogy to the popular interest rate swap options. Due to this similarity, they are also known in the market as **CDS Swaptions**. Swaption products developed to enable trading of credit default risk in the forward space where the default protection might be required in the future. Being optional, CDS Swaptions represent the right, but not the obligation to enter into a CDS contract at pre-defined rate in the future. Since this right is valuable, the option holder pays the option premium. CDS Swaption pricing therefore determines how much the holder needs to pay for such optional forward-starting insurance contract.
Similar to all previous cases, **Mathematica 10** comes as ideal choice for this task as it offers rich probabilistic functionality, powerful integration and elegant differentiation that are all required for this task.

To price credit default option, we need to know the forward value of the CDS.

The value of forward CDS (FCDS) is trivial extension of the standard CDS contract and we utilise all work done in the spot space. Since we have defined the entire hazard rate / IR curve, we can get the forward values of the hazard easily by the following formula : $hf = \frac{1}{t2-t1}\frac{Exp[-h(t1)}{Exp[-h\{t2\}}-1$
And then specify both legs as:
Premium leg: $= \;\; c*N*\sum_t^T Exp[-(z+hf)*i]du$
Protection leg:= $(1-R)*N*h*\sum_t^T Exp[-(z+hf)*i]du$

The meaning of all parameters is the same - c is the premium, N is the nominal amount, z is the risk-free zero –coupon rate, hf is the forward hazard rate (default intensity), R is the recovery rate and dt is the year fraction – generally ¼.

Moving from spot-starting CDS to the forward CDS is simple in Mathematica. One just needs to define different start and end point in integration or summation in the discrete case:

```
fpremleg[x_, t_, y_, n_] :=
  x * Sum[n * f * Exp[-(tdr["PathFunction"][i] + haztd["PathFunction"][i]) * i], {i, t, t + y, f}] // Quiet

fsevleg[t_, y_, n_] :=
   (1 - R) * Sum[n * f * haztd["PathFunction"][t + y] * Exp[-(tdr["PathFunction"][i] + haztd["PathFunction"][i]) * i],
     {i, t, t + y, f}] // Quiet;

FCDS[t_, y_] := fsevleg[t, y, 1] / fpremleg[1, t, y, 1];
```

Although the CDS and FCDS share many commonalities, they behave differently given the time lag in the forward contract. The will generally have a different shape:

1) CDS contract:

```
sw1 = ListLinePlot[Table[{i, CDS[i]}, {i, f, 3, f}],
  PlotStyle → Blue, PlotTheme → "Web", PlotLabel → "Spot CDS"]
```



2) FCDS contract:

```
sw2 = ListLinePlot[Table[{i, FCDS[1, i]}, {i, 1, 4, f}],
  PlotStyle → Red, PlotTheme → "Web", PlotLabel → "Forward CDS"]
```



There is one fundamental difference between IR and credit swaption. Since the exercise of the option means entry into a new IRS (CDS), in credit space this will make sense only if the reference entity is still 'active', i.e. not defaulted. If the entity defaults during the life of the option, the CDS swaption knocks out.

Consider a European *call option* on the CDS rate with nominal =1 where we can buy protection on a reference entity between times t, T for a fixed strike (rate) K. The value of this option is simply $V_0 = A_0 E_Q[Max(F_T - K, 0]$ where A_0 is the value of the 'risky annuity' at time=0, covering the entire product time span E_Q is the risk-neutral expectation with martingale setting and F_T is the forward value of the FCDS.

To value this option we need a model for F_T. A standard assumption (analogy to the IR world) is to assume that F_T is lognormal with zero drift and volatility of $\ln F_T = \sigma\sqrt{T}$. This will generate a well-known Black formula for the forward CDS rate:
We can use the Mathematica's GBM process and get the option value using the *Expectation* function of the payoff formula:

```
gbmproc = GeometricBrownianMotionProcess[0, σ, x0];

gbmev = A0 Expectation[If[x[t] > k, x[t] - k, 0], x ≈ gbmproc, Assumptions → {k > 0, t > 0}] // Simplify
```

$$\frac{1}{2}\left(-k + x0 + k\,\mathrm{Erf}\left[\frac{t\,\sigma^2 + 2\,\mathrm{Log}[k] - 2\,\mathrm{Log}[x0]}{2\sqrt{2}\,\sqrt{t}\,\sigma}\right] + x0\,\mathrm{Erf}\left[\frac{t\,\sigma^2 - 2\,\mathrm{Log}[k] + 2\,\mathrm{Log}[x0]}{2\sqrt{2}\,\sqrt{t}\,\sigma}\right]\right) A_0$$

Two lines of code produce complete analytical result for this option. Numerical value can be obtained by substitution.

Consider 3Y CDS swaption on 2Y CDS with the fixed strike $K = 1.4\%$ forward CDS rate $x0 = 1.4\%$ and the CDS volatility $\sigma = 28\%$ using the curves defined earlier. The option premium =1.22%.

```
gbmev /. {t → 3, x0 → 0.014, k → 0.014, σ → 0.28, μ → 0, A0 → rann}
```

```
0.0122312
```

Analytical solution is useful for a quick sensitivities calculation:

1) Delta = $\frac{\partial V}{\partial x0}$

```
D[gbmev, x0] // Simplify
```

$$\frac{1}{2}\left(1+\mathrm{Erf}\left[\frac{t\,\sigma^2 - 2\,\mathrm{Log}[k] + 2\,\mathrm{Log}[x0]}{2\sqrt{2}\,\sqrt{t}\,\sigma}\right]\right) A_0$$

2) Vega = $\frac{\partial V}{\partial \sigma}$

```
D[gbmev, σ] // Simplify
```

$$\frac{e^{-\frac{t^2\sigma^4 + 4\,\mathrm{Log}[k]^2 - 8\,\mathrm{Log}[k]\,\mathrm{Log}[x0] + 4\,\mathrm{Log}[x0]^2}{8\,t\,\sigma^2}}\sqrt{k}\,\sqrt{t}\,\sqrt{x0}\,A_0}{\sqrt{2\pi}}$$

We can easily display the premium as a function of the FCDS and its volatility

```
Plot3D[gbmev /. {t → 3, k → 0.014, μ → 0, A0 → rann},
 {x0, 0.01, 0.018}, {σ, 0.12, 0.3},
 ColorFunction → "SolarColors", PlotLegends → Automatic]
```



With Mathematica we can go beyond the 'standard' Black swaption model. A suitable alternative is use the so-called **Levy stochastic processes** which accounts for jumps in the intensity regime (hazard rate). **Gamma process** is well-represented Levy process suitable for the credit risk modelling.

This is a typical representation of CDS path simulation with the *Gamma process*:

```
ListLinePlot[Table[(1 - R) * Accumulate[RandomVariate[
   GammaDistribution[4 / 50, 0.005], 50]], {20}],
 PlotLabel → "CDS rates modelled by Gamma Process"]
```



To value an option with the Gamma process setting is simple with Mathematica:

1) Define Gamma distribution
2) Parameterise it to the model:
   a) Drift = $x_0 + \mu t$
   b) Volatility = $\sigma\sqrt{t}$
3) Get the call option expectation with martingale setting $\mu = 0$

```
gdist = GammaDistribution[a, b];
```

```
s1 = Solve[{Mean[gdist] == x0 + μ * t, StandardDeviation[gdist] == σ √t}, {a, b}]
```

$$\left\{\left\{a \to \frac{(x0 + t\,\mu)^2}{t\,\sigma^2},\ b \to \frac{t\,\sigma^2}{x0 + t\,\mu}\right\}\right\}$$

```
ev = A0 Expectation[If[x > k, x - k, 0], x ≈ gdist,
   Assumptions → {k > 0, t > 0}] // Simplify
```

$$\boxed{\frac{\left(-k\,\mathrm{Gamma}\left[a, \frac{k}{b}\right] + b\,\mathrm{Gamma}\left[1+a, \frac{k}{b}\right]\right) A_0}{\mathrm{Gamma}[a]}}$$

```
ev2 = ev /. {a → aval[[1]], b → bval[[1]]}
```

$$\boxed{\frac{\left(\frac{t\,\sigma^2\,\mathrm{Gamma}\left[1+\frac{(x0+t\mu)^2}{t\sigma^2}, \frac{k\,(x0+t\mu)}{t\sigma^2}\right]}{x0+t\mu} - k\,\mathrm{Gamma}\left[\frac{(x0+t\mu)^2}{t\sigma^2}, \frac{k\,(x0+t\mu)}{t\sigma^2}\right]\right) A_0}{\mathrm{Gamma}\left[\frac{(x0+t\mu)^2}{t\sigma^2}\right]}}$$

Gamma model generates surprisingly simple formula and because of embedded jump dynamics, it results in a higher option premium:

```
ev2 /. {t → 3, x0 → 0.014, k → 0.014, σ → 0.00415, μ → 0, A0 → rann}
```

$\boxed{\texttt{0.0127924}}$

Other modelling assumptions for CDS Swaptions include other Levy class – the so-called **Inverse Gaussian** model [IG]. We parameterise the model to the same underlying process:

```
igdist = InverseGaussianDistribution[a, b];
```

```
s12 = Solve[{Mean[igdist] == x0 + μ * t, StandardDeviation[igdist] == σ √t}, {a, b}]
```

$$\left\{\left\{a \to x0 + t\,\mu,\ b \to \frac{(x0 + t\,\mu)^3}{t\,\sigma^2}\right\}\right\}$$

The IG model again produces elegant analytical call option result:

```
s12vals = {a /. s12[[1]], b /. s12[[1]]};
```

```
evig = A0 Expectation[If[x > k, x - k, 0], x ≈ igdist, Assumptions → k > 0 && t > 0] // Simplify
```

$$\boxed{\frac{1}{2}\left(a - k + (a-k)\,\mathrm{Erf}\left[\frac{(a-k)\sqrt{\frac{b}{k}}}{\sqrt{2}\,a}\right] + e^{\frac{2b}{a}}(a+k)\,\mathrm{Erfc}\left[\frac{\sqrt{\frac{b}{k}}\,(a+k)}{\sqrt{2}\,a}\right]\right) A_0}$$

The option premium is lower than Gamma, but higher than LogNormal:

```
calevig /. {t → 3, x0 → 0.014, k → 0.014, σ → 0.00415, μ → 0, A0 → rann}
```

$\boxed{\texttt{0.0123451}}$

With Mathematica we can step into another level and define even more complex model – say **Gamma Process** <u>with non-deterministic volatility</u> . In our example we assume the scale parameter *b* to follow exponential process with parameter λ

```
pm1 = ParameterMixtureDistribution[GammaDistribution[a, b], b ≈ ExponentialDistribution[λ]];
```

```
s15 = Solve[{Mean[pm1] == x0 + μ * t, StandardDeviation[pm1] == σ √t}, {a, λ}]
```

$$\left\{\left\{a \to -\frac{2\,(x0 + t\,\mu)^2}{x0^2 + 2\,t\,x0\,\mu + t^2\mu^2 - t\,\sigma^2},\ \lambda \to -\frac{2\,(x0 + t\,\mu)}{x0^2 + 2\,t\,x0\,\mu + t^2\mu^2 - t\,\sigma^2}\right\}\right\}$$

Although this is quite complex setting, we still manage to get a nice analytical result – in terms of Bessel and Hypergeometric functions which Mathematica evaluates easily:

```
evpm1 = A0 Expectation[If[x > k, x - k, 0], x ≈ pm1, Assumptions → k > 0 && t > 0] // Simplify
```

$$\boxed{\begin{aligned}&-\frac{1}{(1+a)\sqrt{\lambda}\,\mathrm{Gamma}[a]}\\&\mathrm{Csc}[a\pi]\Big(k^{\frac{1+a}{2}}\lambda^{a/2}(k\lambda)^{\frac{1+a}{2}}\mathrm{Gamma}[1-a]\,\mathrm{HypergeometricPFQ}[\{1+a\},\{a, 2+a\}, k\lambda]\,\mathrm{Sin}[a\pi] + (1+a)\Big(k^{\frac{1+a}{2}}\pi\lambda^{a/2}\,\mathrm{BesselI}\big[\\&\quad -1-a, 2\sqrt{k\lambda}\big] - k^{\frac{1+a}{2}}\pi\lambda^{a/2}\sqrt{k\lambda}\,\mathrm{BesselI}\big[-a, 2\sqrt{k\lambda}\big] + 2k\sqrt{\lambda}\,(k\lambda)^{a/2}\,\mathrm{BesselK}\big[-a, 2\sqrt{k\lambda}\big]\,\mathrm{Sin}[a\pi]\Big)\Big) A_0\end{aligned}}$$

After substitution, we see even higher option premium. Given the non-deterministic nature of the volatility, this is reasonably expected:

```
evpmcal /. {t → 3, x0 → 0.014, k → 0.014, σ → 0.004, μ → 0, A0 → rann}
```

$\boxed{\texttt{0.0133833}}$

<u>In short</u>: straightforward setting that make CDS swaptions working just in few steps.